
Changelog
    


  
    
  
  v0.5.0 (2025-07-08)



  
    
  
  Enhancements


	EctoFoundationDB.CLI [doc]: Functions for an operator to use to safely manage
data migrations that cannot be done automatically.
	Index Metadata [doc] now makes use of FoundationDB's \xff/metadataVersion key,
which allows the client to cache metadata and maintain transactional isolation
without having to wait on any keys. Also, the cache is shared across all open tenants of the same id on a given node.
	EctoFoundationDB.Versionstamp [doc]: Added the ability to insert objects with a monotonically increasing integer id, via FoundationDB's versionstamp.
	SchemaMetadata [doc]: This is a new built-in Indexer that allows your app to watch and sync collection of objects in a tenant.
	Added ability to drop an index.


  
    
  
  Breaking changes


	Key construction has changed, making databases created on <= 0.4.x incompatible with >=0.5. Specifically, a binary, atom, or
number primary key is now encoded in the FDB key with the Tuple layer. All other types remain encoded with term_to_binary.
If you need help upgrading your database, please put in a GitHub Issue. We strive for a stable v1.0.


  
    
  
  Bug fixes


	Fixed a bug where index creation was failing for multikey objects
	(#57) Fixed a bug where index management was failing while a new index was being created


  
    
  
  Deprecations


	Ecto has deprecated Repo.transaction in favor of Repo.transact. Since this decision doesn't align with FoundationDB's view of
transactions, we have chosen to deprecate Repo.transaction in favor of Repo.transactional. This terminology better aligns with
:erlfdb and provides a distinction from RDBMS transactions, and allows us to avoid future deprecations.


  
    
  
  New documentation


	Guide for Operators: Describes how to use the EctoFoundationDB.CLI functions to rename a field while guaraneeting that all
concurrent queries in your distributed application are successful.
	Metadata Design: Describes how index metadata is managed and cached in EctoFDB.
	Sync Engine Part I - Single Object: Revamped Livebook that demonstrates how to create a Sync Engine for a single object (for syncing reads)
	Sync Engine Part II - Collections: New Livebook that demonstrates a Sync Engine for a collection of objects in a tenant (still for reads)


  
    
  
  v0.4.0 (2025-01-16)



  
    
  
  Enhancements


	Large Structs: If your struct encodes to a size larger than 100,000 Bytes, it will now be split across several FDB key-values automatically.
Previously, EctoFDB did not attempt to detect this and allowed FoundationDB to throw error code 2103: "value_too_large - Value length exceeds limit".
See documentation for configuration options.
	EctoFoundationDB.Sandbox now uses m:erlfdb_sandbox. Sandbox directory name is now .erlfdb_sandbox. Directories named .erlfdb should be removed.


  
    
  
  Bug fixes


	Upgrade erlfdb to v0.2.2


  
    
  
  v0.3.1 (2024-10-24)



  
    
  
  Bug fixes


	Fixed consistency issue with index updates. Previously, the old index key was still queryable.
	Fixed write amplification issue when updating struct's non-indexed fields.


  
    
  
  New Documentation


	Added fdb_api_counting_text.exs which tests and documents the :erlfdb operations that our Layer is expected to make.


  
    
  
  v0.3.0 (2024-10-20)



  
    
  
  ** Major breaking changes **


Databases that have been created using a prior version of EctoFoundationDB will be broken on
EctoFDB v0.3 and above. Please start a new database with EctoFDB v0.3. If you currently have
a database on v0.2 or earlier, please submit an issue
to discuss the upgrade path.

  
    
  
  Enhancements


	Watches: Support for FDB Watches, which is like a database-driven PubSub on an Ecto struct.
	Directory Tenants: A new default backend for Multitenancy that is production-ready. Managed tenants have been moved to "Experimental" status.
	@schema_context usetenant: true is no longer required.
	The :open_db option now defines a 1-arity function that accepts the Repo module.


  
    
  
  New Documentation


	Livebook | Watches in LiveView



  
    
  
  v0.2.1 (2024-09-23)



  
    
  
  Bug fixes


	Upgrade erlfdb to 0.2.1, allowing Livebook Desktop to discover fdbcli location in /usr/local/bin


  
    
  
  v0.2.0 (2024-09-21)



  
    
  
  Bug fixes



  
    
  
  Enhancements


	Upserts: Support for Ecto options :on_conflict and :conflict_target
	Pipelining: New Repo functions for async/await within a transaction.


  
    
  
  New Documentation


	testing.md: Document to describe how to set up a Sandbox
	CHANGELOG.md: This file!


  
    
  
  v0.1.2 (2024-08-31)



  
    
  
  Bug fixes


	Upgrade erlfdb


  
    
  
  v0.1.1 (2024-08-25)



  
    
  
  Enhancements


	Upgrade to Ecto 3.12


  
    
  
  New documentation


	Livebook | Getting Started: How to get started with EctoFoundationDB.



  
    
  
  v0.1.0 (2024-04-07)



  
    
  
  Features


	Multitenancy
	Basic CRUD operations
	Indexes



  

    
Introduction
    

Mix.install([
  {:ecto_foundationdb, "~> 0.3"}
])

  
    
  
  Setup


Hello! This guide simulates what your experience might be when developing an application with EctoFoundationDB. Specifically, it focuses on the mechanism that EctoFoundationDB uses to create and manage indexes.
It assumes the reader is familiar with general Ecto features.
Before we get started, a couple of important points about executing these commands on your system.
If you received an error on the Mix.install setup, please make sure you have both foundationdb-server and foundationdb-clients packages installed on your system. Also, ensure that your Livebook PATH environment variable includes the directory containing the fdbcli binary.

This LiveBook expects your system to have a running instance of FoundationDB, and it writes and deletes data from it. If your system's /etc/foundationdb/fdb.cluster is pointing to a real database, do not execute these commands!

With that out of the way, we'll start off with creating your Repo module.
defmodule MyApp.Repo do
  use Ecto.Repo, otp_app: :my_app, adapter: Ecto.Adapters.FoundationDB

  use EctoFoundationDB.Migrator

  @impl true
  def migrations() do
    [
      # {0, IndexesMigration}
    ]
  end
end
Notice that the line with IndexesMigration is commented out. We'll come back to this later.

  
    
  
  Developing your app


This next step simulates your app's startup. Normally, you would have a project defining :my_app and the Repo would be included in your supervision tree. In this Guide, we're starting the Repo as an isolated resource.
{:ok, _} = Ecto.Adapters.FoundationDB.ensure_all_started(MyApp.Repo.config(), :temporary)
MyApp.Repo.start_link(log: false)
Next, we define an Ecto.Schema for events that are coming from a temperature sensor. This is a pretty standard Schema module.
defmodule TemperatureEvent do
  use Ecto.Schema

  @primary_key {:id, :binary_id, autogenerate: true}

  schema "temperature_events" do
    field(:recorded_at, :naive_datetime_usec)
    field(:kelvin, :float)
    field(:site, :string)
    timestamps()
  end
end
We're going to create a module that will help us insert some TemperatureEvents.
defmodule Sensor do
  alias Ecto.Adapters.FoundationDB

  def record(n, tenant) do
    MyApp.Repo.transactional(tenant,
      fn ->
        for _ <- 1..n, do: record(nil)
      end)
  end

  def record(tenant) do
    %TemperatureEvent{
      site: "surface",
      kelvin: 373.15 + :rand.normal(0, 5),
      recorded_at: NaiveDateTime.utc_now()
    }
    |> FoundationDB.usetenant(tenant)
    |> MyApp.Repo.insert!()
  end
end
Now, we create and open a new Tenant to store our TemperatureEvents.
alias EctoFoundationDB.Tenant

tenant = Tenant.open!(MyApp.Repo, "experiment-42c")
We're ready to record an event from our temperature sensor! Feel free to Reevaluate this block several times. You'll record 4 new events each time.
for _ <- 1..4, do: Sensor.record(tenant)
We can list all the events from the Tenant. This uses a single FoundationDB Transaction.
MyApp.Repo.all(TemperatureEvent, prefix: tenant)
If there's a large number of events, you can stream them instead of reading them all at once. This uses multiple FoundationDB Transactions.
MyApp.Repo.stream(TemperatureEvent, prefix: tenant)
|> Enum.to_list()
|> length()
Next, we'd like to read all events from "surface". If you're executing this LiveBook in order, you'll receive an exception on this step.
import Ecto.Query

query = from(e in TemperatureEvent, where: e.site == ^"surface")
MyApp.Repo.all(query, prefix: tenant)
Did you get an exception? If so, scroll back up to the defmodule MyApp.Repo block in the Setup section, un-comment the line with IndexesMigration, and Reevaluate that block. Then come back and continue from here. You don't need to Reevaluate other blocks above this text.
👋
Welcome back! You've instructed the Repo to load a migration next time we open a Tenant. But we still need to define that Migration. The block below defines two indexes.
defmodule IndexesMigration do
  use EctoFoundationDB.Migration

  @impl true
  def change() do
    [
      create(index(TemperatureEvent, [:site])),
      create(index(TemperatureEvent, [:recorded_at]))
    ]
  end
end
Now, we re-open the Tenant. Something very important happens here.
This block simulates you restarting your app, and your client reconnecting. We'll just simply call open!/2 again.
tenant = Tenant.open!(MyApp.Repo, "experiment-42c")
Great! If you made it to this step, then the Migration has executed automatically, and the indexes are ready to be used.

  
    
  
  Querying your data


This next block has the same query as the one that threw an exception earlier. This time, you should retrieve the expected events.
import Ecto.Query
query = from(e in TemperatureEvent, where: e.site == ^"surface")
MyApp.Repo.all(query, prefix: tenant)
We can also use the timestamp index that we created in a new query.
now = NaiveDateTime.utc_now()
past = NaiveDateTime.add(now, -1200, :second)

query =
  from(e in TemperatureEvent,
    where: e.recorded_at >= ^past and e.recorded_at < ^now
  )

MyApp.Repo.all(query, prefix: tenant)
Finally, just for fun, let's insert 10,000 TemperatureEvents!
num = 10000
batch = 100

{t, :ok} =
  :timer.tc(fn ->
    Stream.duplicate(batch, div(num, batch))
    |> Task.async_stream(
      Sensor,
      :record,
      [tenant],
      max_concurrency: System.schedulers_online() * 8,
      ordered: false,
      timeout: 30000
    )
    |> Stream.run()
  end)

IO.puts("Done in #{t / 1000} msec")

  
    
  
  Cleaning up


And if you'd like to tidy up, you can easily delete all the data.
# Note: destructive!
MyApp.Repo.delete_all(TemperatureEvent, prefix: tenant)
# Note: destructive!
Tenant.clear_delete!(MyApp.Repo, "experiment-42c")


  

    
Sync Engine Part I - Single Object
    

Mix.install([
  {:phoenix_playground, "~> 0.1.0"},
  {:ecto_foundationdb, github: "foundationdb-beam/ecto_foundationdb"}
])

  
    
  
  Intro


Using FoundationDB Watches, we can set up a mini read-path Sync Engine. It will automatically propagate new data to all mounted LiveViews with push-messaging delivered from the database directly to the LiveView process.
In this demo, we'll update our "favorite quote" in the database and discuss the mechanism used to update the front-end with new data.

  
    
  
  Setup Ecto


First, we set up Ecto, defining a Quote schema, and starting the Repo. For this to work, you must have foundationdb-server installed. Refer to the EctoFoundationDB documentation for installation help.
defmodule Repo do
  use Ecto.Repo, otp_app: :my_app, adapter: Ecto.Adapters.FoundationDB

  use EctoFoundationDB.Migrator

  @impl true
  def migrations(), do: []
end

defmodule Quote do
  use Ecto.Schema

  alias Ecto.Changeset

  @primary_key {:id, :binary_id, autogenerate: true}

  schema "quotes" do
    field(:author, :string)
    field(:content, :string)
    field(:likes, :integer, default: 0)
    field(:as_of, :naive_datetime)
    timestamps()
  end

  def changeset(quote, params \\ %{}) do
    quote
    |> Changeset.cast(params, [:author, :content, :likes, :as_of])
  end

  def like!(tenant, id) do
      Repo.transactional(tenant, fn ->
        quote = Repo.get!(Quote, id)

        quote
        |> changeset(%{likes: quote.likes+1})
        |> Repo.update!()
      end)
  end
end

# ------------------------------------------------------------------
# This section contains some set-up within Livebook that's typically
# handled instead in your application startup and config.
Application.put_env(:my_app, Repo,
  open_db: &EctoFoundationDB.Sandbox.open_db/1,
  storage_id: EctoFoundationDB.Sandbox
)

{:ok, _} = Ecto.Adapters.FoundationDB.ensure_all_started(Repo.config(), :temporary)
Repo.start_link(log: false)
# ------------------------------------------------------------------

alias EctoFoundationDB.Tenant

tenant = Tenant.open!(Repo, "experiment-with-watches-in-liveview")

Next, we create a Quote in the DB. We'll focus on how this specific Quote is rendered throughout the rest of the Livebook.
Repo.insert!(
  %Quote{
    id: "my-favorite-quote",
    author: "Philippe Verdoux",
    content: """
      Enlightenment leads to benightedness; Science entails nescience.
      """,
    as_of: NaiveDateTime.utc_now(:second)
  }, prefix: tenant, on_conflict: :replace_all)

  
    
  
  Setup LiveView


We're using Phoenix Playground to create a sample LiveView. The single code block below packs a bunch of things together. Here are the key takeways:
	In sync_one!/4, we're initializing our mini sync engine.
We do a standard get on the Quote with id "my-favorite-quote" that we've inserted above.
In the same transaction, we're creating a watch with label: :quote.
Finally, we store the :quote and the list of :futures.

	handle_ready/2 is a LiveView hook that receives the :ready message from the watch future, which is delivered directly from the database layer. We use Repo.assign_ready/3 to update the LiveView assigns map. Whenever "my-favorite-quote" changes in the DB, this function will be called automatically. Also, we provide watch?: true to continue listening for updates.

	handle_event/3 receives the "like" event, and uses an FDB transaction to add 1 to the count of likes. Notice we choose not to update the assigns, for demonstration purposes.

	sync_one!/4 and handle_ready/2 both contain nothing specific to the Quote schema. This is a general purpose approach that can be re-used for any other schema types.


defmodule DemoLive do
  use Phoenix.LiveView

  def mount(_params, _session, socket) do

    tenant = Tenant.open!(Repo, "experiment-with-watches-in-liveview")

    {:ok, socket
      |> assign(tenant: tenant)
      |> sync_one!(Quote, :quote, "my-favorite-quote")
    }
  end

  # 1. Initialize our mini Sync Engine
  defp sync_one!(socket, schema, label, id) do
    {struct, futures} = Repo.transactional(socket.assigns.tenant,
      fn ->
        struct = Repo.get!(schema, id)
        future = Repo.watch(struct, label: label)
        {struct, [future]}
      end)

    socket
    |> assign(label, struct)
    |> assign(futures: futures)
    |> attach_hook({:assign_ready, schema, label, id}, :handle_info, &handle_ready/2)
  end

  # 2. LiveView server hook that receives a message from the database layer
  #    and updates the assigns
  defp handle_ready({ref, :ready}, socket) do

    %{assigns: assigns} = socket

    {new_assigns, futures} = Repo.assign_ready(
        assigns.futures,
        [ref],
        watch?: true,
        prefix: assigns.tenant
      )
    {:halt, socket
            |> assign(new_assigns)
            |> assign(futures: futures)}
  end

  def render(assigns) do
    ~H"""
    <div class="content">
      <h1>My Favorite Quote</h1>
      <h6>as of <%= @quote.as_of %> UTC</h6>
      <figure>
        <blockquote><%= @quote.content %></blockquote>
        <h3><figcaption>- <%= @quote.author %></figcaption></h3>

        <p>
          <label>Likes: </label>
          <span><%= @quote.likes %></span>
          <button phx-click="like">👍</button>
        </p>
      </figure>
    </div>

    <style type="text/css">
      h1 { text-align: center; }
      h3 { text-align: right; }
      h6 { text-align: right; }
      p { text-align: center; }

      .content {
        max-width: 500px;
        margin: auto;
      }
    </style>
    """
  end

  # 3. Receive the "like" event and update the DB
  def handle_event("like", _params, socket) do

    Quote.like!(socket.assigns.tenant, socket.assigns.quote.id)

    {:noreply, socket}
  end

end

PhoenixPlayground.start(live: DemoLive)
With the block above evaluated, you can now open a web browser to http://localhost:4000. We suggest you keep this open side-by-side with this Livebook if possible, so that you can watch the updates in real-time. You can even open several tabs to ensure that all tabs are indeed in sync.
You can click the Like button and observe that the value changes. Notice that we are not pushing this value to the assigns explicitly. It's being handled automatically in our mini sync engine in handle_info.

  
    
  
  Updating My Favorite Quote


Your favorite quote might change from moment to moment. We'll write to the database with your current favorite quote. Because of the FDB watch, the LiveView will always render the most up-to-date content.
tenant = Tenant.open!(Repo, "experiment-with-watches-in-liveview")

defmodule ChangeFavoriteQuote do
  @id "my-favorite-quote"

  def to(tenant, author, content) do
    Repo.transactional(tenant, fn ->
      Repo.get(Quote, @id)
      |> Quote.changeset(%{
        author: author,
        content: content,
        as_of: NaiveDateTime.utc_now(:second),
        likes: 0
      })
      |> Repo.update()
      end)
  end
end
ChangeFavoriteQuote.to(
  tenant,
  "Duke Leto Atreides",
  """
  Once you have explored a fear, it becomes less terrifying. \
  Part of courage comes from extending our knowledge.
  """)
ChangeFavoriteQuote.to(tenant, "Captain America", "I can do this all day.")

  
    
  
  Takeaways


The user of our web app is always kept up-to-date with the most recent favorite quote via the combined power of LiveView and our database. In some Phoenix applications, this is done via Phoenix.PubSub, but to do so requires setting up a separate PubSub resource and handling the messages for each kind. In this demo we instead used a feature from FoundationDB, and in doing so, we created a mini Sync Engine. We used a self-contained and generalizable approach composed of the EctoFoundationDB transaction and the LiveView server hook.
The main takeaway from this Livebook is that there is no visible dispatcher. Each process of our application can choose to register a watch, and it will be notified directly as needed from the database itself.


  

    
Sync Engine Part II - Collections
    

Mix.install([
  {:phoenix_playground, "~> 0.1.0"},
  {:ecto_foundationdb, github: "foundationdb-beam/ecto_foundationdb"}
])

  
    
  
  Intro


Using FoundationDB Watches, we can set up a mini read-path Sync Engine. It will automatically propagate new data to all mounted LiveViews with push-messaging delivered from the database directly to the LiveView process.
This is Part II, where we explore syncing a collection of objects to the LiveView page. See Part I for syncing a single object.

  
    
  
  Setup Ecto


First, we set up Ecto, defining a Quote schema, and starting the Repo.
This is similar to Part I, but there's a key difference: we create a SchemaMetadata index on Quote. This metadata is required to drive watches on the collection of quotes.
defmodule Quote do
  use Ecto.Schema

  alias Ecto.Changeset

  @primary_key {:id, :binary_id, autogenerate: true}

  schema "quotes" do
    field(:author, :string)
    field(:content, :string)
    field(:likes, :integer, default: 0)
    field(:as_of, :naive_datetime)
    timestamps()
  end

  def changeset(quote, params \\ %{}) do
    quote
    |> Changeset.cast(params, [:author, :content, :likes, :as_of])
  end

  def like!(tenant, id) do
      Repo.transactional(tenant, fn ->
        quote = Repo.get!(Quote, id)

        quote
        |> changeset(%{likes: quote.likes+1})
        |> Repo.update!()
      end)
  end
end

defmodule QuoteMetadata do
  use EctoFoundationDB.Migration

  @impl true
  def change() do
    [create(metadata(Quote))]
  end
end

defmodule Repo do
  use Ecto.Repo, otp_app: :my_app, adapter: Ecto.Adapters.FoundationDB

  use EctoFoundationDB.Migrator

  @impl true
  def migrations(), do: [
    {1, QuoteMetadata}
  ]
end

# ------------------------------------------------------------------
# This section contains some set-up within Livebook that's typically
# handled instead in your application startup and config.
Application.put_env(:my_app, Repo,
  open_db: &EctoFoundationDB.Sandbox.open_db/1,
  storage_id: EctoFoundationDB.Sandbox
)

{:ok, _} = Ecto.Adapters.FoundationDB.ensure_all_started(Repo.config(), :temporary)
Repo.start_link(log: false)
# ------------------------------------------------------------------

alias EctoFoundationDB.Tenant

tenant = Tenant.open!(Repo, "experiment-with-watches-in-liveview-ii")

Next, we create two Quotes in the DB. Our LiveView page will render the full list.
quotes = [
  {"René Descartes", "Cogito, ergo sum."},
  {"Heraclitus", "One cannot step twice in the same river."}
]

for {author, content} <- quotes, do:
  Repo.insert!(%Quote{author: author, content: content}, prefix: tenant)

  
    
  
  Setup LiveView


Just like in Part I, we're using Phoenix Playground to create a sample LiveView. Here are the key takeways:
	In sync_all!/4, we're initializing our mini sync engine. Notice that we use Repo.all and SchemaMetadata.watch_changes.

	handle_ready/2 is the same LiveView hook as Part I.

	sync_all!/4 and handle_ready/2 both contain nothing specific to the Quote schema. This is a general purpose approach that can be re-used for any other schema types.


defmodule DemoLive do
  use Phoenix.LiveView

  alias EctoFoundationDB.Indexer.SchemaMetadata

  def mount(_params, _session, socket) do

    tenant = Tenant.open!(Repo, "experiment-with-watches-in-liveview-ii")

    {:ok, socket
      |> assign(tenant: tenant)
      |> sync_all!(Quote, :quotes)
    }
  end

  # 1. Initialize our mini Sync Engine. Similar to Part I, except this time we
  #    use Repo.all and SchemaMetadata.watch_changes
  defp sync_all!(socket, schema, label) do
    {list, futures} = Repo.transactional(socket.assigns.tenant,
      fn ->
        list = Repo.all(schema)
        future = SchemaMetadata.watch_changes(schema, label: label)
        {list, [future]}
      end)

    socket
    |> assign(label, list)
    |> assign(futures: futures)
    |> attach_hook({:assign_ready, schema, label}, :handle_info, &handle_ready/2)
  end

  # 2. LiveView server hook that receives a message from the database layer
  #    and updates the assigns (same as Part I)
  defp handle_ready({ref, :ready}, socket) do

    %{assigns: assigns} = socket

    {new_assigns, futures} = Repo.assign_ready(
        assigns.futures,
        [ref],
        watch?: true,
        prefix: assigns.tenant
      )
    {:halt, socket
            |> assign(new_assigns)
            |> assign(futures: futures)}
  end

  def render(assigns) do
    ~H"""
    <div class="content">
      <h1>A list of Quotes</h1>
      <ul>
        <li :for={quote <- @quotes}>
          <figure>
            <blockquote><%= quote.content %></blockquote>
            <h3><figcaption>- <%= quote.author %></figcaption></h3>

            <p>
              <label>Likes: </label>
              <span><%= quote.likes %></span>
              <button phx-click="like" phx-value-id={quote.id}>👍</button>
            </p>
          </figure>
        </li>
      </ul>
    </div>

    <style type="text/css">
      h1 { text-align: center; }
      h3 { text-align: right; }
      h6 { text-align: right; }
      p { text-align: center; }
      ul li { list-style-type: none; }

      .content {
        max-width: 500px;
        margin: auto;
      }
    </style>
    """
  end

  # 3. Receive the "like" event and update the DB
  def handle_event("like", params, socket) do

    Quote.like!(socket.assigns.tenant, params["id"])

    {:noreply, socket}
  end

end

PhoenixPlayground.start(live: DemoLive)
Open a web browser to http://localhost:4000 side-by-side with this Livebook, so that you can watch the updates in real-time.
You can click the Like button and observe that the value changes. Notice that we are not pushing this value to the assigns explicitly. It's being handled automatically in our mini sync engine in handle_info. This is all wired together because of our use of watch_changes on the Quote schema. There are various functions that notify your app in different circumstances:
	watch_inserts: Notifies when an insert or upsert happens
	watch_deletes: Notifies when a delete happens
	watch_collection: Notifies when either an insert or a delete happens
	watch_updates: Notifies when an update (Repo.update) happens
	watch_changes: Notifies when an insert, delete, or update happens

All of these notifications are scoped to the tenant.

  
    
  
  Inserting and deleting some Quotes


To demonstrate that the page will update as expected, we can insert and delete some Quote objects while we have the page open.
tenant = Tenant.open!(Repo, "experiment-with-watches-in-liveview-ii")

defmodule CollectQuotes do
  def insert(tenant, author, content) do
    Repo.insert!(%Quote{author: author, content: content}, prefix: tenant)
  end

  def delete_random(tenant) do
    Repo.transactional(tenant, fn ->
      Repo.all(Quote)
      |> Enum.random()
      |> Repo.delete!()
    end)
  end
end
CollectQuotes.insert(tenant, "Elixir", "Hello World")
CollectQuotes.delete_random(tenant)

  
    
  
  Takeaways


In Part I, we showed how EctoFDB can implement a mini Sync Engine for the read-path of a single object in the database. In Part II, we upgraded it to a collection of objects on the tenant.
These two approaches can be combined easily - the LiveView hook we defined works for both types at the same time. The only difference is how you initialize on mount.


  

    
Testing with ExUnit
    

Setting up your app to use a FoundationDB Sandbox is very easy!
First, in config/test.exs:
config :my_app, MyApp.Repo,
  open_db: &EctoFoundationDB.Sandbox.open_db/1
Next, set up an ExUnit Case that will provide you with a new Tenant to use
in your tests. This step is optional, but recommended: it will make writing
each test frictionless.
defmodule MyApp.TenantCase do
  @moduledoc false
  use ExUnit.CaseTemplate

  alias EctoFoundationDB.Sandbox

  setup do
    tenant_id = Ecto.UUID.autogenerate()

    tenant = Sandbox.checkout(MyApp.Repo, tenant_id, [])

    on_exit(fn ->
      Sandbox.checkin(MyApp.Repo, tenant_id)
    end)

    {:ok, [tenant_id: tenant_id, tenant: tenant]}
  end
end
Now, you can use your TenantCase to do any FDB operation in a test. Because
we're using tenants with randomized names, you don't have to worry about key conflicts, so you can include the async: true option.
defmodule MyAppHelloFDBTest do
  use MyApp.TenantCase, async: true

  alias Ecto.Adapters.FoundationDB

  test "hello", context do
    tenant = context[:tenant]

    # An :erlfdb hello world, delete this for your tests
    assert :not_found ==
             FoundationDB.transactional(
               tenant,
               fn tx ->
                 tx
                 |> :erlfdb.get("hello world")
                 |> :erlfdb.wait()
               end
             )

    # Example:
    #   MyApp.Repo.insert!(%MyApp.Hello{message: "world"}, prefix: tenant)
  end
end


  

    
Operator's Manual
    

In order to maintain transactional guarantees as your application changes and
grows, there is a burden on the database operator for some data manipulations.
EctoFoundationDB provides building blocks that
can be executed at strategic times to maintain data integrity.
There is a need for a human operator to execute these commands for one
important reason: EctoFDB does not know details about the topology of your
distributed application, or the migration versions that are active on clients
at a given time.
In writing this guide, we've taken heavy inspiration from
GitHub | Safe Ecto Migrations
It's written assuming a SQL database like Postgres, but many of the concepts
are applicable to EctoFoundationDB, and it's worth a read.
Safe Index Migrations
Adding and dropping indexes is done completely on-line without risk of data consistency problems.
The remainder of this document discusses other data operations. Please
refer to the documentation on
Migrations
for more info on index migrations.


  
    
  
  Safe Data Migrations


	Adding a field
	Removing a field
	Renaming a field


  
    
  
  Adding a field


Adding a field is trivial. Simply modify your schema to include the new field.
There is no ALTER TABLE DDL or equivalent. Any existing objects will
automatically interpret values for the new field as nil. If you require
some other value, use the Repo.update!/2 function like normal.
# App deploy, in the Ecto schema

defmodule MyApp.Post do
  schema "posts" do
+   field :a_new_field, :string
  end
end


  
    
  
  Removing a field


If all you care about is not having the field show up in your structs, you may
choose to simply remove it from your schema. The underlying data will remain
intact, however, until each item is updated at some point in the future. If you
need to ensure that the data is truly removed, continue reading.
Safety can be assured if the application code is first updated to remove references to the field so it's no longer loaded or queried. Then, the field can safely be removed from the database.
	Deploy code change to remove references to the field.
	Execute EctoFoundationDB.CLI.delete_field!/3 to remove the data.

Warning: data rewrite
Removing a field requires writing all keys that have that field in the
value.

Application deployment:
# App deploy, in the Ecto schema

defmodule MyApp.Post do
  schema "posts" do
-   field :no_longer_needed, :string
  end
end
Once your application is fully deployed to all nodes, execute the following
command:
iex> EctoFoundationDB.CLI.delete_field!(MyApp.Repo, MyApp.Post, :no_longer_needed)


  
    
  
  Renaming a field


Take a phased approach:
	Create a new field with any relevant indexes
	In application code, write to both fields
	Backfill data from old field to new field (see below)
	In application code, move reads from old field to the new field
	In application code, remove old field from Ecto schemas, and drop any old index(es)
	Delete the old field (see below)

Warning: data rewrite
Renaming a field requires writing all related keys twice.

Backfill data from old field to new field
iex> EctoFoundationDB.CLI.copy_field!(MyApp.Repo, MyApp.Post, :old_field, :new_field)
Delete the old field
iex> EctoFoundationDB.CLI.delete_field!(MyApp.Repo, MyApp.Post, :old_field)
Sample Available
Please refer to GitHub | cli_test.exs
to review the test that executes this operation.



  

    
Metadata
    

EctoFDB stores metadata alongside your Ecto schema data in order to maintain
correct data integrity on the indexes. In doing so, each EctoFDB client
knows which indexes are needed to be written to, even if the application code
is on an older version, which can be the case when deploying updates to a
distributed system.
Since this metadata is required to be available in every transaction, EctoFDB
must allow for the retrieval of the metadata in every transaction. However,
we wish to avoid that retrieval from actually happening as much as possible.
Otherwise, those particular keys become very "hot" and limit the scalability
of the system. Therefore, we maintain a local cache of the metadata for each
(tenant, schema) tuple.
This document describes the design considerations for the metadata and the
approaches we use to ensure the cache is guaranteed to be valid.

  
    
  
  Metadata Content


We currently keep track of 2 types of metadata: indexes and partial indexes.

  
    
  
  Indexes


This is a list of all 'ready' indexes for a particular Schema. An index is
considered 'ready' when it has been created and is ready to be used.
At query time, this list of indexes is inspected to determine the index to use
to provide the best possible query performance.
At insert and update time, the list is used to set or clear the index keys
according to the specified EctoFoundationDB.Indexer.

  
    
  
  Partial Indexes


This is a list of all indexes that are currently undergoing a migration,
usually for index creation. The creation of an index for a particular
(tenant, schema) tuple can take arbitrarily long, and so any queries that
arrive in the meantime must be handled accordingly. The list of partial
indexes informs EctoFDB of how to keep data integrity for any concurrent
creation of Default indexes.

  
    
  
  Metadata Cache


The cache uses 2 stages for invalidation.
	Stage 1: The global FoundationDB metadata version key
	Stage 2: A version key for each (tenant, schema) tuple, and a special
"claim key" for any partial indexes

Metadata Version Key
This is a key that has special treatment in the implementation of
FoundationDB transactions. The
FDB Design Doc | Metadata Version
provides a detailed explanation of how this key is implemented.
For the purposes of this document, it's important to understand that the metadata version key is truly global to the entire keyspace. Since we allow
tenants to migrate independently, this necessarily means that the migration for
a single tenant will invalidate Stage 1 of the cache for all tenants in
the  database.
There is no cost to reading the global metadata version key, since it's
always sent along with other necessary transactional data. This means that
at steady state, your transactions only need to do the base minimum FDB
operations, and are able to avoid any hot keys.
If the global version is found to have changed, we do not yet invalidate the
cache. Instead, we move onto Stage 2 of the cache invalidation process. In
doing so, we limit the impact of another tenant's migration.
Schema Migration Version Key
This is a key that is managed by EctoFDB itself, and does not have special
treatment by FDB. It always contains an integer value representing the largest
version number for complete migrations that are specified in your
EctoFoundationDB.Migrator.
When the global key is found to have changed in Stage 1 of the cache
invalidation, we perform a 'get' on this key. Then we delay the wait until
after some of the transaction work has been completed. Thus, the cache
is optimistic that the cached value is still valid. Only at the end of the
transaction do we wait for and compare the version value. If it's not equal,
then the cache is invalidated and the transaction is retried.
Claim Key
This key is used for tracking metadata that is currently undergoing a
migration. It's managed in the same way as the schema migration version key.
Its content includes a cursor that defines the progress of the migration.


  

    
Ecto.Adapters.FoundationDB 
    



      
(Hex.pm | GitHub)
Adapter module for FoundationDB.
It uses erlfdb for communicating with the database,
and implements an Ecto-compatible Layer on top of FDB's key-value store.

  
    
  
  Installation


Install the latest stable release of FoundationDB from the
official FoundationDB Releases.
The foundationdb-server package is required on any system that will be running
a FoundationDB server instance. For example, it's common to
run the foundationdb-server on your development machine and on managed
instances running a FoundationDB cluster, but not for your stateless Elixir
application server in production.
foundationdb-clients is always required.
Include :ecto_foundationdb in your list of dependencies in mix.exs:
defp deps do
  [
    {:ecto_foundationdb, "~> 0.5"}
  ]
end

  
    
  
  Usage


Define this module in your MyApp application:
defmodule MyApp.Repo do
  use Ecto.Repo, otp_app: :my_app, adapter: Ecto.Adapters.FoundationDB

  use EctoFoundationDB.Migrator

  @impl true
  def migrations(), do: []
end
Edit your config.exs file to include the following:
config :my_app, MyApp.Repo,
  cluster_file: "/etc/foundationdb/fdb.cluster"
In your app's supervisor, you can start the repo as follows:
children = [
  ...
  MyApp.Repo,
  ...
]

Supervisor.start_link(...)

  
    
  
  Tenants


EctoFoundationDB requires your application to use multitenancy via EctoFoundationDB.Tenant.
We strongly encourage your application to decide on a multitenancy strategy that makes sense
for your problem domain (or simply use "default" to get started).
Once you have opened a tenant, you'll use the Ecto :prefix option to provide the tenant
to each Repo operation.
Retrieving data from a tenant yields no execution overhead. In effect, each tenant can be
viewed as a logically partitioned database. See EctoFoundationDB.Tenant for a discussion
on the mechanism used to achieve this in FDB.

  
    
  
  Creating a schema


This is a standard Ecto.Schema that will be used in the examples of this documentation.
defmodule User do
  use Ecto.Schema
  @primary_key {:id, :binary_id, autogenerate: true}
  schema "users" do
    field(:name, :string)
    field(:department, :string)
    field(:balance, :integer)
    timestamps()
  end
  # ...
end

  
    
  
  Opening a tenant


See EctoFoundationDB.Tenant for more details.
alias EctoFoundationDB.Tenant

tenant = Tenant.open!(MyApp.Repo, "some-org")

  
    
  
  Inserting a new struct


The Tenant is stored on the struct as metadata, so that when passing structs
to Repo functions, the :prefix option is not required.
alice = %User{name: "Alice", department: "Engineering"}
       |> FoundationDB.usetenant(tenant)

alice = MyApp.Repo.insert!(alice)

  
    
  
  Querying for a struct using the primary key


The :prefix option is required. The struct returned will have the tenant
in the metadata, like above.
MyApp.Repo.get!(User, alice.id, prefix: tenant)

  
    
  
  Transaction basics


Here's a simple example of transaction usage. Notice that the calls to Repo inside the transaction function
do not include the :prefix option. See Transactions for more information.
MyApp.Repo.transactional(tenant, fn ->
  alice = MyApp.Repo.get!(User, alice.id)
  if alice.balance > 0 do
    MyApp.Repo.update(User.changeset(alice, %{balance: alice.balance - 1}))
  else
    raise "Overdraft"
  end
end)

  
    
  
  Indexes and queries


The implication of the FoundationDB Layer Concept
is that the manner in which data can be stored and accessed is the responsibility of the client
application. To achieve general purpose data storage and access patterns commonly desired
in web applications, EctoFoundationDB provides support for indexes out of the box, and
also allows your application to define custom indexes.
When you define a migration that calls create index(User, [:department]), a Default index is
created on the :department field. The following two indexes are equivalent:
# ...
create index(User, [:department])
# is equivalent to
create index(User, [:department], options: [indexer: EctoFoundationDB.Indexer.Default]))]
# ...
Timestamp indexes and multi-field indexes are supported:
create index(User, [:inserted_at])
create index(User, [:birthyear, :department, :birthdate])
In fact, the index value can be any Elixir term. All types support Equal queries (of the form x.field == "value").
Certain Ecto Schema types support Between queries (of the form x.field >= 10 and x.field < 20).
See the Data Types section for the list.
See the Migrations section for further details about managing indexes in your application.

  
    
  
  Index query examples


Retrieve all users in the Engineering department (with an Equal constraint):
query = from(u in User, where: u.department == ^"Engineering")
MyApp.Repo.all(query, prefix: tenant)
Retrieve all users with a birthyear from 1992 to 1994 (with a Between constraint):
query = from(u in User,
  where: u.birthyear >= ^1992 and u.birthyear <= ^1994
)
MyApp.Repo.all(query, prefix: tenant)
Retrieve all Engineers born in August 1992:
query = from(u in User,
  where: u.birthyear == ^1992 and
         u.department == ^"Engineering" and
         u.birthdate >= ^~D[1992-08-01] and u.birthdate < ^~D[1992-09-01]
)
MyApp.Repo.all(query, prefix: tenant)

  
    
  
  Index ordering matters


When you create an index using multiple fields, the FDB key that stores the index will be extended with
all the values in the order of your defined index fields. Because FoundationDB stores the keys themselves
in a well-defined order, the order of the fields in your index determines the Between queries you can perform.
RULE:  For any Query, there can be 0 or 1 Between clauses, and if one exists for a field, then there must not
exist an Equal clause for ay field that follows, according to the list of fields in the index. Moreover, the Between
field must be of a Data Type supported for Between queries.

Above, we demonstrated a nontrivial query with 2 Equal constraints and 1 Between constraint. Compare that with an invalid
query below, which attempts a Between constraint on the :birthyear field followed by an Equal constraint on the :department
field.
# Invalid query!
from(u in User,
  where: u.birthyear >= ^1992 and u.birthyear < ^1995 and
         u.department == ^"Engineering"
)
Why is this query unsupported by EctoFDB? With the index defined as-is, EctoFDB is unable to extract a minimal
dataset from the database in one operation. Instead of over-extracting data and then applying a filter, we've chosen
to reject the query. We hope this encourages correct index management and gives the developer confidence about the
expected execution time for any given call to Repo.
Yeah that's cool, but how do I run the query I want to run? You have 2 options.
	Purposefully over-extract the data using Stream, and do the filtering in Elixir:

from(u in User, where: u.birthyear >= ^1992 and u.birthyear < 1995 and)
|> MyApp.Repo.stream(prefix: tenant)
|> Stream.filter(&((&1).department == "Engineering"))
|> Enum.to_list()
or
	Create the necessary index. For our example above, the new index might look like:

create index(User, [:department, :birthyear])
We encourage you to test execution times of your expected workload before deciding to create an index.
Remember: since your data is already partitioned into tenants, you're effectively getting 1 index for free.

  
    
  
  Custom indexes


As data is inserted, updated, deleted, and queried, the Indexer callbacks (via behaviour EctoFoundationDB.Indexer)
are executed. Your Indexer module does all the work necessary to maintain the index within the transaction
that the data itself is being manipulated. Many different types of indexes are possible. You are in control!
For an example, please see the Default implementation.

  
    
  
  Transactions


All work involving FDB happens in a transaction, including each Repo function call (e.g. Repo.get/1). EctoFoundationDB also
exposes the Repo.transactional/2 function, allowing you to group operations together with transactional isolation (ACID
and globally serializable)
MyApp.Repo.transactional(tenant, fn ->
    # Ecto work
  end)
Please read the FoundationDB Developer Guide on transactions
for more information about how to develop with transactions.
It's important to remember that even though the database gives ACID guarantees about the
keys updated in a transaction, the Elixir function passed into Repo.transactional/2
may be executed more than once when any conflicts
are encountered. For this reason, your function must not have any side effects other than
the database operations. For example, do not publish any messages to other processes
(such as Phoenix.PubSub) from within the transaction.
# Do not do this!
MyApp.Repo.transactional(tenant, fn ->
  MyApp.Repo.insert(%User{name: "Alice"})
  ...
  Phoenix.PubSub.broadcast("my_topic", "new_user") # Not safe! :(
end)
# Instead, do this:
MyApp.Repo.transactional(tenant, fn ->
  MyApp.Repo.insert(%User{name: "Alice"})
  ...
end)
Phoenix.PubSub.broadcast("my_topic", "new_user") # Safe :)
Note: If you're looking for PubSub-like functionality pushed from the database itself, please see
Watches.

  
    
  
  Migrations


At first glance, EctoFoundationDB migrations may look similar to that of :ecto_sql,
but the actual execution of migrations and how your app must be configured are very
different, so please read this section in full.
If your app uses indexes on any of your schemas, you must include a module that implements
the EctoFoundationDB.Migrator behaviour. By default, this is assumed to be your Repo
module, as we have demonstrated above.
Migrations are only used for index management. There are no tables to add, drop, or modify.
When you add a field to your Schema, any read requests on old objects will return nil in
that field.

  
    
  
  Online migrations


As tenants are opened during your application runtime, migrations will be executed
automatically in a "lazy" fashion. This distributes the migration across a potentially long period of time,
as migrations will not be executed unless the tenant is opened. Specifically, a call to
Tenant.open/2 will block until the migration for that tenant completes. Other tenants
will not be affected.
For example, imagine you have 2 nodes running your application. Node 1 has opened tenants "a"
and "b". You deploy a new version of your app to Node 2 that includes a new index, and open
tenant "a" there. On Node 2 only, that call to Tenant.open/2 will block until the migration
completes. All operations on Node 1 will remain working. When Tenant.open/2 completes on Node 2
for tenant "a", the index is available to all "a" tenants across all nodes. Any "b" tenants
remain unaffected throughout this process.

  
    
  
  Configuring your app


The Migrator is a module in your application runtime that provides the full list of
ordered migrations. These are the migrations that will be executed when a tenant is opened.
If you leave out a migration from the list, it will not be applied.
The result of migrations/0 will change over time to include new migrations, like so:
defmodule MyApp.Repo do
  # ...
  def migrations() do
    [
      {0, MyApp.IndexUserByDepartment}
      # At a later date, we my add a new index with a corresponding addition
      # tho the migrations list:
      #    {1, MyApp.IndexUserByRole}
    ]
  end
end
Each migration is contained in a separate module, much like EctoSQL's. However, the operations
to be carried out must be returned as a list. For example, the creation an index
may look like this:
defmodule MyApp.IndexUserByDepartment do
  use EctoFoundationDB.Migration

  @impl true
  def change() do
    [
      create index(User, [:department])
    ]
  end
end
Your migrator and MyApp.<Migration> modules are part of your codebase. They must be
included in your application release. This differs from traditional Ecto migrations (:ecto_sql), which
are typically managed in .exs scripts executed outside the production application runtime.

  
    
  
  Creating and dropping indexes


In your Migration, to create an index use create index(table, columns) and to drop an index use drop index(table, columns).
Indexes are created and dropped in a manner that is safe for your application transactions.
defmodule MyApp.ExampleMigration do
  use EctoFoundationDB.Migration

  @impl true
  def change() do
    [
      create index(User, [:role]),
      drop index(User, [:department]),
    ]
  end
end

  
    
  
  Executing all migrations immediately


Migrations can be completed in full at any time with a call to
EctoFoundationDB.CLI.migrate!/1, which accepts your repo module (e.g. MyApp.Repo).
# May take a while to complete
EctoFoundationDB.CLI.migrate!(MyApp.Repo)
Depending on how many tenants you have in your database, and the size of the data for each tenant,
this operation might take a long time and make heavy use of system resources. It is safe to interrupt
this operation, as migrations are performed transactionally. However, if you interrupt a migration, the next
attempt for that tenant may have a brief delay (~5 sec) while the migrator ensures that the
previous migration has indeed stopped executing.
Please refer to the documentation of EctoFoundationDB.CLI for other actions to be carried out
by a human operator.

  
    
  
  Data types


EctoFoundationDB stores your struct's data using :erlang.term_to_binary/1, and
retrieves it with :erlang.binary_to_term/1. As such, there is no data type
conversion between Elixir types and Database Types. Any term you put in your
struct will be stored and later retrieved.
Data types are used by EctoFoundationDB for the creation and querying of indexes.
Certain Ecto types are encoded into the FDB key, which allows you to formulate
Between queries on indexes of these types.
 Between queries are supported for these types:
	:id
	:binary_id
	:integer
	:float
	:boolean
	:string
	:binary
	:date
	:time
	:time_usec
	:naive_datetime
	:naive_datetime_usec
	:utc_datetime
	:utc_datetime_usec

See Indexes and Queries for more information.

  
    
  
  Versionstamps (autoincrement)


FoundationDB provides a mechanism for generating an automatically increasing integer id, called a versionstamp.
Versionstamps are guaranteed to be unique and increasing along with the total order of the serializable transactions.
They are not guaranteed to increment by exactly one.
To use Versionstamps, your Schema must have a primary key field of type Versionstamp, with autogenerate: false.
@primary_key {:id, EctoFoundationDB.Versionstamp, autogenerate: false}
The easiest way to insert such records is by using Repo.async_insert_all/3. This function follows a different
approach than the other Repo.async_* functions: the future must be awaited outside of the transaction.
alias MyApp.{Repo, Event}
future = Repo.transactional(tenant, fn ->
  Repo.async_insert_all(Event, [%Event{data: "event_a"}, %Event{data: "event_b"}])
end)
[event_a, event_b] = Repo.await(future)
The purpose of the async/await pattern here is for the FDB client to discover the committed versionstamp, and to apply
it to the inserted structs after the transaction is committed. Since the computation of the versionstamp is a
commit-time decision by the FoundationDB Server, the final value is unknowable by the client until after the transaction
is committed.
Inspecting the contents of the :id field will reveal the structure of the versionstamp:
event_a.id
# => {:versionstamp, 1111, 2222, 0}
The Versionstamp can be converted to an integer using the EctoFoundationDB.Versionstamp.to_integer/1 function.
alias EctoFoundationDB.Versionstamp
Versionstamp.to_integer(event_a.id)
# => 4771854286848
Note:
	The approach to inserting above is guaranteed to have zero conflicts with other keys in the database.
	A group of records inserted in a single transaction will have versionatmps that are guaranteed to increment by 1.
	Records inserted in different transactions will not have a predictable versionstamp distance from each other. That distance
is very unlikely to ever be 1.


  
    
  
  Watches


FoundationDB Watches are
similar to Triggers in an RDBMS. Registering a watch on a particular key provides a guarantee*
that when that key in the database is changed, the client application is notified with a
push-style notification. This registration is initiated inside a transaction, and the caller is
provided a EctoFoundationDB.Future object that is used to receive the push notification at a later time.
The notification is delivered directly to the Elixir process that requested it.
The EctoFDB Repo module allows you to register a watch on any struct that you have written
to the database. The FDB watch is registered on the key that represents the Primary Write (the key-value
that stores all the data of your struct). When that struct changes in the database via any means,
the Future is resolved and the caller has the opportunity to refresh its internal state.

  
    
  
  EctoFDB Repo functions for watches


EctoFDB adds the following functions to your Repo module. They are each discussed in detail
below.
	Repo.watch/1/Repo.watch/2: Registers a watch
	Repo.assign_ready/2/Repo.assign_ready/3: Handles Future resolution via a push message


  
    
  
  Repo.watch/2


With arguments struct and options, registers
a watch on the given struct and returns a Future. This Future resolves to
a value when the FDB key is set or cleared. Until then, the Future remains unresolved.
The Future from a watch is resolvable outside of an FDB transaction.
In addition to other standard options, a :label option will prime the Future for use by assign_ready/3.
It's recommended that you provide a label and use assign_ready/3, but not required.

  
    
  
  Repo.assign_ready/3


With arguments futures, ready_refs, and
options, returns a tuple containing: 1) a Keyword of structs with keys according to the :label
provided to watch/2, and 2) an updated list of futures that are still unresolved.
In order for the caller to acquire the ready_refs arguments, you must receive messages from the
process mailbox of the form {ref, :ready}. See the example below for detail. If your process chooses
to buffer a list ready_refs longer than 1, then assign_ready/3 uses pipelining to retrieve the
resulting structs. Each watched struct must have been registered with a unique :label.
The options are provided to Repo.async_get/3 internally. In addition to standard options, a
watch?: true option will re-register a new watch on the struct so that your calling
process's event loop can always be aware of changes.

  
    
  
  Example


The power of watches is best illustrated in the context of a stateful Elixir process that is
interested in tracking the up-to-date value for a particular schema. This could be a GenServer, LiveView,
or some other stateful process.
defmodule MyAliceView do

  # ...

  def init(tenant) do
    {alice, future} = MyRepo.transactional(tenant,
                        fn ->
                          alice = MyRepo.get_by(User, name: "Alice")
                          {alice, MyRepo.watch(alice, label: :alice)}
                        end)
    assigns = [alice: alice]
    {:ok, %{assigns: assigns, futures: [future]}}
  end

  # ...

  def handle_info({ref, :ready}, state=%{assigns: assigns, futures: futures, tenant: tenant}) when is_reference(ref) do
    {new_assigns, new_futures} = MyRepo.assign_ready(futures, [ref], watch?: true, prefix: tenant)
    {:noreply, %{
          state |
            assigns: Map.merge(assigns, Enum.into(new_assigns, %{})),
            futures: new_futures
          }}
  end
end
By handling the Future message resolution event {ref, :ready} in the handle_info/2, we never need to
call Repo.await/1. Instead, our process can do other work and react to the :ready event immediately
when it's delivered.
And Repo.assign_ready/3 provides a conventional way to retrieve the result and register a new watch
in one go, so that updates to Alice are not missed.
For more on watches, see
	Sync Engine Part I - Single Object
	Schema Metadata


  
    
  
  Schema Metadata


In addition to Default Indexes, EctoFDB provides another built-in Indexer called EctoFoundationDB.Indexer.SchemaMetadata.
This is an optional Indexer that will keep track of various actions for a Schema:
	inserts: Incremented for each insert or upsert
	deletes: Incremented for each delete
	collection: Incremented for each insert, update, or delete
	updates: Incremented for each update (via Repo.update/*)
	changes: Incremented for each insert, upsert, delete, or update

These keys are useful for creating watches that will notify your application of those actions. For example, if you create a watch on the inserts key,
your application will be notified when a new record is inserted, and you can react however you like.
The counters are scoped to a particular tenant. If you desire a different scoping for the watches, then we recommend you split your tenant.
To create this Indexer, you use Migrations:
defmodule MyApp.UserSchemaMetadata do
  @moduledoc false
  use EctoFoundationDB.Migration

  @impl true
  def change() do
    [create metadata(User)]
  end
end
  def migrations() do
    [
      # ...
      {2, MyApp.UserSchemaMetadata},
      # ...
    ]
  end
For more on these watches, see Sync Engine Part II - Collections

  
    
  
  Upserts


The FoundationDB Adapter supports the following upsert approaches.

  
    
  
  Upserts with on_conflict


The following choices for on_conflict are supported, and they work in the manner
described by the official Ecto documentation.
	:raise: The default.
	:nothing
	:replace_all
	{:replace_all_except, fields}
	{:replace, fields}


  
    
  
  Upserts with conflict_target


If you provide conflict_target: [], the FoundationDB Adapter will make no effort to
inspect the existence of objects in the database before insert. When an existing
object is blindly overwritten, your indexes may become inconsistent, resulting in undefined
behavior. You should only consider its use if one of the following is true:
	(a) your schema has no indexes
	(b) you know apriori that any indexed fields are unchanged at the time of upsert, or
	(c) you know apirori that the primary key you're inserting doesn't already exist.

For these use cases, this option may speed up the loading of initial data at scale.

  
    
  
  Pipelining


Pipelining is the technique of sending multiple queries to a database at the same time, and
then receiving the results at a later time. In doing so, you can often avoid waiting
for multiple network round trips. EctoFDB uses a async/await syntax to
implement pipelining.

  
    
  
  EctoFDB Repo functions for pipelining


EctoFDB adds the following functions to your Repo module. You'll notice that there are async_*
versions of all the Queryable functions on the standard Ecto.Repo. These have the same arguments
as their counterparts, and they each return a single EctoFoundationDB.Future.
	Repo.async_get/2 / Repo.async_get/3
	Repo.async_get!/2 / Repo.async_get!/3
	Repo.async_get_by/2 / Repo.async_get_by/3
	Repo.async_get_by!/2 / Repo.async_get_by!/3
	Repo.async_one/1 / Repo.async_one/2
	Repo.async_one!/1 / Repo.async_one!/2
	Repo.async_all/1 / Repo.async_all/2
	Repo.await/1: The results of the above functions are provided to Repo.await/1 to
resolve the Futures. It accepts a single Future or a list of Futures.


  
    
  
  Example


Consider this example. Our transaction below has 2 network round trips in serial.
The "Alice" User is retrieved, and then the "Bob" user is retrieved. Finally, the
list is constructed and the transaction ends.
result = MyApp.Repo.transactional(tenant, fn ->
  [
    MyApp.Repo.get_by(User, name: "Alice"),
    MyApp.Repo.get_by(User, name: "Bob")
  ]
end)

[alice, bob] = result
With pipelining, we can issue 2 async_get_by queries without waiting for their result,
and then only when we need the data, we await the result. Notice that the list of futures
is constructed, and then we await on them.
result = MyApp.Repo.transactional(tenant, fn ->
  futures = [
    MyApp.Repo.async_get_by(User, name: "Alice"),
    MyApp.Repo.async_get_by(User, name: "Bob")
  ]

  MyApp.Repo.await(futures)
end)

[alice, bob] = result
Conceptually, only 1 network round-trip is required, so this will be faster than
waiting on each in sequence.

  
    
  
  Pipelining on range queries (interleaving waits)


The functions Repo.async_all/1 and Repo.async_all/2 work a little bit different on the await. When the DB must
return a list of results, if that list ends up being larger than :erlfdb's configured :target_bytes,
then the result set is split into multiple round-trips to the database. If your application awaits multiple
such queries, then :erlfdb's "interleaving waits" feature is used.
Consider this example, where we're retrieving all Users and all Posts in the same transaction.
MyApp.Repo.transactional(tenant, fn ->
  futures = [
    MyApp.Repo.async_all(User),
    MyApp.Repo.async_all(Post)
  ]

  MyApp.Repo.await(futures)
end)
Each query individually executed would be multiple round-trips to the database as pages of results are retrieved.
sequenceDiagram
    Client->>Server: get_range(User)
    Server-->>Client: {100 users, continuation_token}
    Client->>Server: get_range(User, continuation_token)
    Server-->>Client: {10 users, done}
With :erlfdb's interleaving waits, we send as many get_range requests as possible with pipelining
until all are exhausted.
sequenceDiagram
    Client->>Server: get_range(User)
    Client->>Server: get_range(Post)
    Server-->>Client: {100 users, continuation_token}
    Server-->>Client: {100 posts, continuation_token}
    Client->>Server: get_range(User, continuation_token)
    Client->>Server: get_range(Post, continuation_token)
    Server-->>Client: {10 users, done}
    Server-->>Client: {100 posts, continuation_token}
    Client->>Server: get_range(Post, continuation_token)
    Server-->>Client: {2 posts, done}
This happens automatically when you include the list of futures to your Repo.await/1.

  
    
  
  Standard options


	:cluster_file - The path to the fdb.cluster file. The default is
"/etc/foundationdb/fdb.cluster".
	:migrator - A module that implements the EctoFoundationDB.Migrator
behaviour. Required when using any indexes. Defaults to nil. When nil,
your Repo module is assumed to be the Migrator.


  
    
  
  Advanced options


	:storage_id - All tenants created by this adapter are prefixed with
this string. This allows multiple configurations to operate on the
same FoundationDB cluster independently. Defaults to
"Ecto.Adapters.FoundationDB".
	:open_db - 1-arity function accepting the Repo module. Used for opening a reference to the
 FoundationDB cluster. Defaults to &EctoFoundationDB.Database.open/1. When
 using EctoFoundationDB.Sandbox, you should set this option to Sandbox.open_db/1.
	:migration_step - The maximum number of keys to process in a single transaction
when running migrations. Defaults to 1000. If you use a number that is
too large, the FDB transactions run by the Migrator will fail.
	:metadata_cache - When set to :enabled, the Ecto ets cache is used to store the
available indexes per tenant. This speeds up all database operations.
Defaults to :enabled.
	:max_single_value_size - Number of Bytes above which an encoded struct will
be split into multiple FDB key-value pairs upon insert and update. Defaults to 100_000.
Any value beyond 100_000 is unsupported by FoundationDB. You probably do not
want to change this value. Also see :max_value_size.
	:max_value_size - Number of Bytes above which an encoded struct will be
rejected by EctoFDB upon insert and update. Defaults to :infinity, which
means that a single encoded struct of any size can be written to any number of
keys. Please remember that FDB's transaction size limit of 10MB still applies.
The maximum number of key-value pairs for a given encoded struct, not counting
indexes, is computed by 1 + ceil(:max_value_size / :max_single_value_size).


  
    
  
  Optimizing for throughput and latency


Pipelining operations inside your transactions can offer significant speed-ups, and you should
also consider parallelizing multiple transactions. Both techniques are useful in different
circumstances.
	Pipelining: The transaction can contain arbitrary logic and remain ACID compliant. If you make
smart use of pipelining, the latency of execution of that logic is kept to a minimum.
	Parallel transactions: Each transaction is internally ACID compliant, but any 2 given
transactions may conflict if they're accessing the same keys. When transactions do not conflict,
throughput goes up.

If you're attempting to get the maximum throughput on data loading, you'll probably want to
make use of both!
Read more at FDB Developer Guide | Loading Data.

  
    
  
  Limitations and caveats



  
    
  
  FoundationDB


Please read FoundationDB Known Limitations.

  
    
  
  Layer


EctoFoundationDB implements a specific Layer on the FoundationDB
key-value store. This Layer is intended to be generally useful, but you
may not find it suitable for your workloads. The Layer is documented in
detail at EctoFoundationDB.Layer. This project does not support
plugging in other Layers.

  
    
  
  Tenants


	The use of a tenant implies an ownership relationship between the tenant and the data.
It's up to you how you use this relationship in your application.

	Failure to specify a tenant will result in a raised exception at runtime.



  
    
  
  Migrations


The following are not supported:
	Renaming a table
	Renaming a field
	Deleting a field GitHub #32
	Dropping an index
	Moving down in migration versions (rollback)

Finally, do not use the  Ecto Mix tasks:
# These commands are not supported. Do not use them with :ecto_foundationdb!
#    mix ecto.create
#    mix ecto.drop
#    mix ecto.migrate
#    mix ecto.gen.migration
#    mix ecto.rollback
#    mix ecto.migrations

  
    
  
  Key and value size


FoundationDB has limits on key and value size
Please be aware of these limitations as you develop. EctoFDB does support splitting an individual struct
across multiple keys in the database. This is automatic and doesn't require any additional input from the developer. The
options :max_single_value_size and :max_value_size allow you to tweak this feature.

  
    
  
  Ecto queries


You'll find that most queries outside of those detailed in the documentation fail with
an EctoFoundationDB.Exception.Unsupported exception. The FoundationDB Layer concept
precludes complex queries from being executed within the database. Therefore, it only makes sense
to implement a limited set of query types -- specifically those that do have optimized database
query semantics. All other filtering, aggregation, and grouping must be done by your Elixir code.
In other words, you'll be writing less SQL, and more Elixir.
Refer to the integration tests for a collection of queries that is known to work.

  
    
  
  Other Ecto features


As you test the boundaries of EctoFoundationDB, you'll find that many other Ecto features just plain
don't work with this adapter. Ecto has a lot of features, and this adapter is in active development.
Please let us know what you find!

    EctoFoundationDB.CLI - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.CLI 
    



      
This module provides functions that are to be run by an operator, and not
part of application code.

    EctoFoundationDB.Database - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.Database 
    



      
See Ecto.Adapters.FoundationDB

    EctoFoundationDB.Future - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.Future 
    



      
Opaque struct that represents an unresolved result set from some FoundationDB
query operation.
If you have received a Future from an EctoFDB Repo API, please consult the documentation
for that API for dealing with the Future. The functions here are intended to be for
internal use only.

    EctoFoundationDB.Index - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.Index 
    



      
See Ecto.Adapters.FoundationDB

      


      
        Summary


  
    Types
  


    
      
        
  
    
    EctoFoundationDB.Indexer - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.Indexer behaviour
    



      
Implement this behaviour to create a custom index.
Each Indexer has access to read, write, and clear any and all data in the database.
A faulty implementation may lead to data loss or corruption.

    EctoFoundationDB.Indexer.SchemaMetadata - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.Indexer.SchemaMetadata 
    



      
This is Indexer keeps track of various actions for a Schema:
	inserts: Incremented for each insert or upsert
	deletes: Incremented for each delete
	collection: Incremented for each insert, update, or delete
	updates: Incremented for each update (via Repo.update/*)
	changes: Incremented for each insert, upsert, delete, or update

These keys are useful for creating watches that will notify your application of those actions. For example, if you create a watch on the inserts key,
your application will be notified when a new record is inserted, and you can react however you like.
See it in action: Sync Engine Part II - Collections

EctoFoundationDB.Layer 
    



      
The Ecto FoundationDB Adapter implements a Layer
on the underlying key-value store provided by FoundationDB. Via this layer, some
common data access patterns are achieved. For those familiar with relational databases
such as Postgres, these patterns will be familiar. However, there are many differences (for example SQL is
not supported), so this document seeks to describe the capabilities of the Ecto FoundationDB Layer in detail.

  
    
  
  Keyspace Design


All keys used by :ecto_foundationdb are encoded with FoundationDB's Tuple Encoding.
With the default :tenant_backend of EctoFoundationDB.Tenant.DirectoryTenant, each tuple written by EctoFDB is
prefixed with a short binary string as allocated by :erlfdb_directory. This tenant prefixing is the most critical element of the keyspace because it's the mechanism that guarantees tenants cannot cross their boundaries. **The rest of this documentation is written assuming you're using the DirectoryTenant.**
The first element of the tuple after the tenant prefix is a string prefix that is intended to keep the
:ecto_foundationdb keyspace separate from other keys in the FoundationDB cluster.
Your Schema data and Default indexes are stored with "\xFD".
The data associated with schema migrations is stored with "\xFE".
The rest of the tenant's keyspace is open for use by you, the application developer. For example, it is safe to write:
db = FoundationDB.open(MyApp.Repo)
tenant = Tenant.open(MyApp.Repo, "some-org")

# Multitenancy-Safe. The key is properly packed
:erlfdb.set(db, Tenant.pack(tenant, {"hello"}), "world")

# Multitenancy-Unsafe. The key is not packed into the tenant's keyspace
:erlfdb.set(db, :erlfdb_tuple.pack({"hello"}), "world")
:erlfdb.set(db, "hello", "world")
A value (the binary stored at each key) is either
	some other keys (in the case of Default indexes) or
	Erlang term data encoded with :erlang.term_to_binary/1


  
    
  
  Primary Write and Read


Your Ecto Schema has a primary key field, which is usually a string or an integer. This primary
key uniquely identifies an object of your schema within the tenant in which it lives.
defmodule EctoFoundationDB.Schemas.User do
  use Ecto.Schema
  schema "users" do
    field(:name, :string)
    field(:department, :string)
    timestamps()
  end
end
In this example, a User has an :id and a :name. Also remember that the User is defined within
a tenant which provides a scope under which the User lives. For example, a typical tenant
would be the organization the User belongs to. Since the User is in this tenant, we do not need
to provide an identifier for this organization on the User object itself.
"Primary Write" refers to the insertion of the User struct into the FoundationDB key-value store
under a single key that uniquely identifies the User. This key includes the :id value.
Your struct data is stored as a Keyword encoded with :erlang.term_to_binary/1.
Note: The Primary Write can be skipped by providing the write_primary: false option on the @schema_context.
See below for more.

  
    
  
  Default Indexes


When a Default index is created via a migration, the Ecto FoundationDB Adapter writes a set of
keys and values to facilitate lookups based on the indexed field.

  
    
  
  Advanced Options: write_primary: false, mapped?: false


If you choose to use write_primary: false on your schema, this skips the Primary Write. The consequence of this are as follows:
	You'll want to make sure your index is created with an option mapped?: false. This ensures that the
struct data is written to the index keyspace.
	Your index queries will now have performance characteristics similar to a primary key query. That is, you'll
be able to retrieve the struct data with a single erlfdb:get_range/3.
	The data can only be managed by providing a query on the index. You will not be able to access the data
via the primary key.
	If write_primary: false, then only one index can be created.


      




  

  
    
    EctoFoundationDB.Migrator - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.Migrator behaviour
    



      
Implement this behaviour to define migrations for Ecto.Adapters.FoundationDB

EctoFoundationDB.Options 
    



      
See Ecto.Adapters.FoundationDB

    EctoFoundationDB.QueryPlan - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.QueryPlan 
    



      
See Ecto.Adapters.FoundationDB

    EctoFoundationDB.Sandbox - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.Sandbox 
    



      
A module for managing a sandbox FoundationDB cluster. This allows a developer to create
a space under which it should be safe to write tests.
When using this module, it creates a directory in the project root called .erlfdb. It
is safe to delete this directory when you no longer need it (e.g. after test execution)
See Testing with EctoFoundationDB for more.

EctoFoundationDB.Tenant 
    



      
This module allows the application to create, open, and delete
tenants within the FoundationDB database. All transactions require a tenant,
so any application that uses the Ecto FoundationDB Adapter must use this module.
EctoFoundationDB supports 2 different backends for the Tenant implementation:
	EctoFoundationDB.Tenant.DirectoryTenant: The default. Tenants are managed via
the :erlfdb_directory layer. Each tenant is assigned a byte prefix. EctoFDB
puts that prefix as the first element of every tuple key. If your application wishes
to write keys directly to the database, it must also respect this prefix.
	EctoFoundationDB.Tenant.ManagedTenant: Uses the experimental FoundationDB Tenants.
This requires tenant_mode=required_experimental or tenant_mode=optional_experimental
to be enabled on your FDB system. EctoFDB does not need to prefix keys because it happens
automatically at the FDB transaction level. If your application wishes to write keys
directory to the database, there is no prefix to worry about. Use caution: FDB Tenants
are not yet tested with the same rigor as the rest of FDB.

If you wish to use the experimental ManagedTenant, add this option to your Repo config:
config :my_app, MyApp.Repo,
  tenant_backend: EctoFoundationDB.Tenant.ManagedTenant
EctoFDB does not support switching between tenant backends on a database. If you want to switch backend,
you must use a new empty database.

    EctoFoundationDB.Tenant.DirectoryTenant - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.Tenant.DirectoryTenant 
    



      
The default backend for EctoFDB multitenancy.
It uses the Directory Layer defined by :erlfdb_directory to
partition tenant keyspaces from each other. There is no guarantee
provided by FDB at the transaction-level, so care must be taken to
always pack keys appropriately. Standard use of Ecto.Repo functions
is guaranteed to respect tenant boundaries.

    EctoFoundationDB.Tenant.ManagedTenant - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.Tenant.ManagedTenant 
    



      
An experimental backend for EctoFDB multitenancy. It uses FDB's Tenants
to guarantee that a given transaction cannot access keys from another tenant.
To use ManagedTenant, your database must be configured with
fdbcli --exec 'configure tenant_mode=required_experimental'
# or
fdbcli --exec 'configure tenant_mode=optional_experimental'

FDB's tenants are still an experimental feature, so EctoFDB's ManagedTenant must
also be considered experimental. For now, we recommend using EctoFoundationDB.Tenant.DirectoryTenant,
which is the default.

EctoFoundationDB.Versionstamp 
    



      
Versionstamping is a feature that allows you to create a unique identifier for a record
that is guaranteed to be unique across all records in the database.
Please refer to the documentation for Repo.async_insert_all!/3.

    EctoFoundationDB.Exception.IncorrectTenancy - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.Exception.IncorrectTenancy exception
    



      
This exception is raised when there is a conflict in the tenant used.

      




  

  
    
    EctoFoundationDB.Exception.Unsupported - Ecto.Adapters.FoundationDB v0.5.0
    
    

    


  
  

    
EctoFoundationDB.Exception.Unsupported exception
    



      
This exception is raised when the application uses an Ecto feature that is not
supported by the FoundationDB Adapter.

